
=====================SIMPLE_PM_WAVEFORM_GENERATION===================================== 
1)  1V_pk_True_PM is nonlinear           = constant 1V_pk signal
2)  1V_pk_100%_AM for this FFT           = 1_pk_carrier and 0.5_pk_sidebands
3)  1V_pk_1_radian_PM                    = 0.816_pk_carrier and 0.408_pk_sidebands
4)  1V_pk_0.1_radian_PM                  = 1_pk_carrier and 0.05_pk_sidebands
5)  FFT_resolution_Hz                    = 1/total_time_sec
6)  Nyqust_Hz                            = 0.5/Sample_time_sec
7)  Num_Bins for FFT                     = Nyqust_Hz/FFT_resolution_Hz

1) A true PM signal does not change the magnitude of a carrier.
   Therefore a +/- one radian signal will produce a nonlinear spectrum.
2) A +/- one radian PM spectrum tries to resemble a 100% AM spectrum with a 1_pk_carrier 
   and two 0.5_pk_sidebands.
3) The PM nonlinearity produces instead a 0.816_pk_carrier and 0.408_pk_sidebands.

=======================Find_Ave_RmsCOS=====================================
RMS_level_Expect       .707   RMS_level_Cos  0.707494  
=======================Find_Ave_RmsPM=====================================
RMS_level_Expect       .707   RMS_level_RM  0.706941  
=======================FFT_and_Plot_PM=====================================
FFT_BandWidth_Hz=      500000
FFT_resolution_Hz=     1000
Fundamental+sideband   0.765163 + 0.44003 + 0.440031 
Total_RMS              0.986269 
=======================FFT_and_Plot_Vcos=====================================
FFT_BandWidth_Hz=      500000
FFT_resolution_Hz=     1000
Fundamental+sideband   0.999954 + 1.47099E-07 + 3.50213E-08 
Total_RMS              0.999954 

4) For ten time smaller modulation, both a 10%_AM and a 10%_radian_PM produce a 1_pk_carrier 
   and two 0.05_pk_sidebands.



=======================Find_Ave_RmsCOS=====================================
RMS_level_Expect       .707   RMS_level_Cos  0.707494  
=======================Find_Ave_RmsPM=====================================
RMS_level_Expect       .707   RMS_level_RM  0.707486  
=======================FFT_and_Plot_PM=====================================
FFT_BandWidth_Hz=      500000
FFT_resolution_Hz=     1000
Fundamental+sideband   0.997456 + 0.0499352 + 0.0499354 
Total_RMS              0.999953 
=======================FFT_and_Plot_Vcos=====================================
FFT_BandWidth_Hz=      500000
FFT_resolution_Hz=     1000
Fundamental+sideband   0.999954 + 1.59501E-07 + 3.57084E-08 
Total_RMS              0.999954 

5) One needs for instance 1 whole second to measure a signal to a 1Hz resolution.
6) At least two samples are needed to detect the presents of an AC signal.
7) FFT translate samples over time to bins of frequency. The bin's width is 
   the FFT_resolution_Hz and the Maximum frequency is Nyqust_Hz

===================MacSpiceCode====================================================
SIMPLE_PM_WAVEFORM_GENERATION
*=========Create_Signal==================
VTime     VTime  0      DC     0    PWL(    0      0    1      1)
Vfreq1    Vfreq1 0      DC     2
BMOD      VMOD   0      V      = cos(6.2831853*2000*V(VTime))    
BPM       VPM    0      V      = 1*cos(6.2831853*100k*V(VTime)+1*V(VMOD))
BCOS      VCOS   0      V      = 1*cos(6.2831853*100k*V(VTime))

.control
*TRAN            TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran             .1u    1m     0       .1u   
set              pensize = 2
plot             vpm    vcos
echo             "=======================Find_Ave_RmsCOS====================================="
let averVal =    mean(vcos)
let noisAC =     vcos - averVal
let RmsVal =     sqrt(mean(noisAC* noisAC))
echo             "RMS_level_Expect       .707   RMS_level_Cos  $&RmsVal  " 
unlet averVal
unlet RmsVal
echo             "=======================Find_Ave_RmsPM====================================="
let averVal =    mean(VPM)
let noisAC =     VPM - averVal
let RmsVal =     sqrt(mean(noisAC* noisAC))
echo             "RMS_level_Expect       .707   RMS_level_RM  $&RmsVal  " 
unlet averVal
unlet RmsVal
echo             "=======================FFT_and_Plot_PM====================================="
linearize
let              FFT_BandWidth_Hz =      500k
let              FFT_resolution_Hz =     1k
echo             "FFT_BandWidth_Hz=      $&FFT_BandWidth_Hz"
echo             "FFT_resolution_Hz=     $&FFT_resolution_Hz"
set              specwindow=   "rectangular"
spec             $&FFT_resolution_Hz  $&FFT_BandWidth_Hz   $&FFT_resolution_Hz     v(vpm)
plot             mag (vpm) ylog xlimit 90k 110k
plot             mag (vpm) ylog xlimit 95k 105k ylimit .1 1
let fund =       mag(vpm[99])
let upsb =       mag(vpm[101])
let lpsb =       mag(vpm[97])
echo             "Fundamental+sideband   $&fund + $&upsb + $&lpsb " 
let totalrms =   sqrt( fund*fund +upsb*upsb+ lpsb*lpsb)
echo             "Total_RMS      $&totalrms " 
echo             "=======================FFT_and_Plot_Vcos====================================="
destroy
let              FFT_BandWidth_Hz =      500k
let              FFT_resolution_Hz =     1k
echo             "FFT_BandWidth_Hz=      $&FFT_BandWidth_Hz"
echo             "FFT_resolution_Hz=     $&FFT_resolution_Hz"
set              specwindow=   "rectangular"
spec             $&FFT_resolution_Hz  $&FFT_BandWidth_Hz   $&FFT_resolution_Hz     v(vcos)
plot             mag (vcos) ylog xlimit 90k 110k
plot             mag (vcos) ylog xlimit 95k 105k ylimit .1 1
let fund =       mag(vcos[99])
let upsb =       mag(vcos[101])



let lpsb =       mag(vcos[97])
echo             "Fundamental+sideband   $&fund + $&upsb + $&lpsb " 
let totalrms =   sqrt( fund*fund +upsb*upsb+ lpsb*lpsb)
echo             "Total_RMS      $&totalrms "

.endc

.end

